```
;; Source-to-sink agent based model for sediment transport in Ancient Lake Bonneville.
;; written by Chelsea Welker for GEOG 6160, Spring 2016



extensions [ gis ]
globals [
  study-area
  sink-area
  source-area
  sed-distance ;; distance of moved particle
  travel-dirN ;; prevailing Southerly winds
  travel-dirS ;; prevailing Northerly winds
  travel-dirSE ;; prevailing Northwesterly winds
  tick-count
  total-time
  total-distance
  total-turtles
  storm-heading
  dem-back
  elevation
]
patches-own [
  is-source-area?
  is-sink-area?
  elev
  ]

turtles-own [
  start-patch
  ]

to setup
  ca

  set total-time 0
  set total-distance 0
  set total-turtles 0

  ;;set elevation gis:load-dataset ("shp/dem98utm83.asc")
  ;;gis:paint elevation 40
  ;;gis:apply-raster elevation elev


  set study-area gis:load-dataset ("shp/model_area.shp") ;;set boundaries of model
  gis:set-world-envelope gis:envelope-of study-area

  set sink-area gis:load-dataset ("shp/sink_area.shp") ;; set boundary of sink area
  gis:set-drawing-color [255 0 0]
  gis:draw sink-area 2

  set source-area gis:load-dataset ("shp/geology_alluv.shp") ;; set boundary of source area

  ask patches [ set is-source-area? false ] ;; set patches equal to source area, blue
  ask patches gis:intersecting source-area
  [ set is-source-area? true
    set pcolor blue]

  ask patches [ set is-sink-area? false ] ;; set patches equal to source area, red
  ask patches gis:intersecting sink-area
  [ set is-sink-area? true
    set pcolor red]


  ask n-of 1000 patches with [pcolor = blue and not any? other turtles-here]
  [sprout 1
    [set size 1
      set color white
      set start-patch patch-here]] ;; create 1000 turtles on blue source areas
```

```
  set sed-distance (sed-velocity * (storm-length * 3600)) * .000001 ;;convert cm to km; d = r * t

  set travel-dirS 75.0 ;; probability 75% coming from N
  set travel-dirSE 75.0 ;; probability 75% coming from NW
  set travel-dirN 10.0 ;; probability 10% coming from S

  reset-ticks

end

to show-patches
  clear-drawing

  ask patches with [ is-source-area? ]
  [
    set pcolor blue
  ]

  ask patches with [ is-sink-area? ]
  [
    set pcolor red
  ]
end


to go
  if ticks >= 1000 [ stop ]

  ifelse random-float 100.0 < travel-dirN
    [set storm-heading 0 + random-float 10]
    [ifelse random-float 100.0 < 50
      [set storm-heading 170 + random-float 20]
      [set storm-heading 125 + random-float 20]
    ]

  ask n-of 500 patches with [pcolor = blue]
  [if not any? other turtles-here
    [sprout 1
      [set size 1
        set color white
        set start-patch patch-here]]] ;; create infinite turtles on blue source areas

  ask turtles
  [
    ifelse pcolor != red
      [ move ]
      [
        if color != yellow
          [ tally ]
      ]
  ]

  tick
end

to recolor-patches
  ask patches [
    ifelse elev >= 0 [ set pcolor scale-color green elev 2000 5000 ] [
      set pcolor blue
    ]
  ]
end

to move
  set heading storm-heading

  if not can-move? sed-distance
    [ die ]
```

```
  fd sed-distance
end


to tally
  set total-time total-time + (( storm-length * ticks ) / 1440) ;; converting to days
  set total-distance total-distance + ( sed-distance * ticks )
  set total-turtles total-turtles + 1
  set color yellow
end


to stay
  ;; do nothing
end


to-report mean-distance ;;calculate mean distance of turtles
  ifelse total-turtles > 0
    [ report total-distance / total-turtles ]
    [ report 0 ]
end


to-report mean-time
  ifelse total-turtles > 0
    [ report total-time / total-turtles ]
    [ report 0 ]
end
```